PRÁCTICA 5: Integrales de Superficie



Integral de superficie de funciones escalares

Sea $f(x,y,z)$ una función continua con valores reales definida en S. Definimos la integral de f sobre S como

$$\begin{aligned}\iint_S f(x,y,z)dS &= \iint_S fdS = \iint_D f\left(\overline{T}(u,v)\right) \|\overline{T}_u \times \overline{T}_v\| \, dudv \\ &= \iint_D f(x(u,v),y(u,v),z(u,v)) \|\overline{T}_u \times \overline{T}_v\| \, dudv\end{aligned}$$

Calcularemos el módulo de un vector $\vec{V}$ como la raiz cuadrada del producto escalar:

$$||\vec{V}|| = \sqrt{\vec{V}\cdot\vec{V}}$$

- Para simplificar expresiones trigonométricas utilizaremos la orden **trigsimp** y para simplificar otras expresiones utilizaremos la orden **ratsimp.**
- Para calcular el producto escalar de dos vectores con *wxMaxima,* utilizaremos el punto del teclado **.**
- Para el producto vectorial utilizaremos la orden **~** (Alt Gr 4 del teclado), cargando previamente el paquete **load(vect).**

```
(%i1)    t1:[1,2,3];
(t1)     [1,2,3]
(%i2)    t2:[a,b,c];
(t2)     [a,b,c]
(%i3)    pe:t1.t2;
(pe)     3 c+2 b+a
(%i4)    load(vect);
(%o4)    C:\maxima-5.38.1\share\maxima
```

```
(%i5)   pv:t1~t2;
(pv)    [1,2,3]~[a,b,c]
```

Como se ve, *wxMaxima* se limita a devolvernos, para el producto vectorial, la misma orden que le introducimos.

- Para que nos de el valor de dicho producto vectorial necesitamos la orden **express**.

```
(%i6)   express(pv);
(%o6)   [2 c-3 b,3 a-c,b-2 a]
```

Ejemplo 1

Calcular la integral de superficie.

$$\iint_S z dS$$

Siendo S la superficie del paraboloide $z = 1 - x^2 - y^2$, situada por encima del plano OXY.

a) Utilizando coordenadas cartesianas.

b) Utilizando coordenadas polares.

Solución con wxMaxima:

```
(%i7)   load(draw)$
0 errors, 0 warnings
(%i8)   wxdraw3d(enhanced3d=true,surface_hide=true,
            implicit(z=1-x^2-y^2,x,-2,2,y,-2,2,z,-1,1));
(%t8)
```




```
(%o8)
```

```
-->     /* a) En coordenadas cartesianas */

-->     /* Ecuaciones paramétricas */
(%i11)  x(u,v):=u;
        y(u,v):=v;
        z(u,v):=1-u^2-v^2;
```

(%o9) $x(u,v):=u$

(%o10) $y(u,v):=v$

(%o11) $z(u,v):=1-u^2-v^2$

```
-->     /* Cálculo de la normal */
(%i16)  t(u,v):=[x(u,v),y(u,v),z(u,v)];
        tu:diff(t(u,v),u);
        tv:diff(t(u,v),v);
        n:tu~tv;
        normal:express(n);
```

(%o12) $t(u,v):=[x(u,v),y(u,v),z(u,v)]$

(tu) $[1,0,-2\,u]$

(tv) $[0,1,-2\,v]$

(n) $-[0,1,-2\,v]\sim[1,0,-2\,u]$

(normal) $[2\,u,2\,v,1]$

```
-->     /* Cálculo del módulo de la normal */
(%i17)  m:sqrt(normal.normal);
```

(m) $\sqrt{4\,v^2+4\,u^2+1}$

```
-->     /* Cálculo de la función integrando */
(%i19)  f(x,y,z):=z;
        int:f(x(u,v),y(u,v),z(u,v))*m;
```

(%o18) $f(x,y,z):=z$

(int) $\left(-v^2-u^2+1\right)\sqrt{4\,v^2+4\,u^2+1}$

```
-->     /* Cálculo de la integral */
(%i20)  iterada1:integrate(int,v,-sqrt(1-u^2),sqrt(1-u^2));
```

(iterada1) $$-\frac{\left(48\,u^4-56\,u^2-17\right)\operatorname{asinh}\left(\frac{2\sqrt{1-u^2}}{\sqrt{4\,u^2+1}}\right)+\sqrt{1-u^2}\left(24\sqrt{5}\,u^2-14\sqrt{5}\right)}{32}$$

```
(%i21)  quad_qags(iterada1, u, -1, 1);
```

(%o21) $[2.351047484775507,3.942801640732796\,10^{-11},399,0]$

```
-->      /* b) En coordenadas polares */

-->      /* Ecuaciones paramétricas */
(%i24)   x(u,v):=u*cos(v);
         y(u,v):=u*sin(v);
         z(u,v):=1-u^2;
```

(%o22) $x(u,v):=u\cos(v)$

(%o23) $y(u,v):=u\sin(v)$

(%o24) $z(u,v):=1-u^2$

```
-->      /* Cálculo de la normal */
(%i29)   t(u,v):=[x(u,v),y(u,v),z(u,v)];
         tu:diff(t(u,v),u);
         tv:diff(t(u,v),v);
         n:tu~tv;
         normal:express(n);
```

(%o25) $t(u,v):=[x(u,v),y(u,v),z(u,v)]$

(tu) $[\cos(v),\sin(v),-2\,u]$

(tv) $[-u\sin(v),u\cos(v),0]$

(n) $[\cos(v),\sin(v),-2\,u]\sim[-u\sin(v),u\cos(v),0$

(normal) $[2\,u^2\cos(v),2\,u^2\sin(v),u\sin(v)^2+u\cos(v)^2]$

```
-->      /* Cálculo del módulo de la normal */
(%i30)   m:sqrt(normal.normal);
```

(m) $\sqrt{\left(u\sin(v)^2+u\cos(v)^2\right)^2+4\,u^4\sin(v)^2+4\,u^4\cos(v)^2}$

```
-->      /* Cálculo de la función integrando */
(%i32)   f(x,y,z):=z;
         int:f(x(u,v),y(u,v),z(u,v))*m;
```

(%o31) $f(x,y,z):=z$

(int) $\left(1-u^2\right)\sqrt{\left(u\sin(v)^2+u\cos(v)^2\right)^2+4\,u^4\sin(v)^2+4\,u^4\cos(v)^2}$

```
(%i33)   int:trigsimp(int);
```

(int) $\left(1-u^2\right)\sqrt{4\,u^4+u^2}$

```
-->      /* Cálculo de la integral */
(%i34)   iterada1:integrate(int,u,0,1);
```

(iterada1) $\dfrac{5^{3/2}}{24}-\dfrac{11}{120}$

```
(%i36) integrate(iterada1, v, 0, 2*%pi);
```

(%o36) $2\left(\frac{5^{3/2}}{24}-\frac{11}{120}\right)\pi$

```
(%i37) %,numer;
(%o37) 2.351047484775508
```

Integrales de superficie de funciones vectoriales

Supongamos que S es una superficie definida por la parametrización

$$\begin{aligned} \overline{T} &: D \subset \mathbb{R}^2 \to S \subset \mathbb{R}^3 \\ \overline{T}(u,v) &= x(u,v)\overline{i} + y(u,v)\overline{j} + z(u,v)\overline{k} \end{aligned}$$

Sea $\overline{F}(x,y,z) = F_1(x,y,z)\overline{i} + F_2(x,y,z)\overline{j} + F_3(x,y,z)\overline{k}$ *un campo vectorial continuo definido en* S*. La integral de superficie de* $\overline{F}$ *sobre* S*, denotada por*

$$\iint_S \overline{F}\cdot\overline{dS}$$

la definimos como

$$\iint_S \overline{F}\cdot\overline{dS} = \iint_D \overline{F}\left(\overline{T}(u,v)\right)\cdot\left(\overline{T}_u \times \overline{T}_v\right) dudv$$

La integral de superficie también se pueden representar en *forma diferencial* por

$$\iint_S \overline{F}\cdot\overline{dS} = \iint_S F_1 dydz + F_2 dxdz + F_3 dxdy$$

Ejemplo 2

Calcular la integral de superficie $\iint_S \overrightarrow{F}.\overrightarrow{dS}$, siendo

$$\overrightarrow{F} = xz^2\,\overrightarrow{i} + (x^2y - z^3)\,\overrightarrow{j} + (2xy + y^2z)\,\overrightarrow{k}$$

y S la superficie de la esfera $x^2+y^2+z^2=1$, situada por encima del plano OXY.
a) Utilizando coordenadas cartesianas.
b) Utilizando coordenadas esféricas.

Solución con wxMaxima:

```
-->      /* a) En coordenadas cartesianas */
-->      /* Ecuaciones paramétricas */
(%i38) x(u,v):=u;
       y(u,v):=v;
       z(u,v):=sqrt(1-u^2-v^2);
```

(%o36) $x(u,v):=u$

(%o37) $y(u,v):=v$

(%o38) $z(u,v):=\sqrt{1-u^2-v^2}$

(%i40)
```
wxdraw3d(enhanced3d=true,surface_hide=true,
implicit(x^2+y^2+z^2=1,x,-1,1,y,-1,1,z,0,1));
```

(%t40)



(%o40)

--> `/* Cálculo de la normal */`

(%i44)
```
t(u,v):=[x(u,v),y(u,v),z(u,v)];
tu:diff(t(u,v),u);
tv:diff(t(u,v),v);
n:tu~tv;
normal:express(n);
```

(%o40) $t(u,v):=[x(u,v),y(u,v),z(u,v)]$

(tu) $[1,0,-\frac{u}{\sqrt{-v^2-u^2+1}}]$

(tv) $[0,1,-\frac{v}{\sqrt{-v^2-u^2+1}}]$

(n) $-[0,1,-\frac{v}{\sqrt{-v^2-u^2+1}}]\sim[1,0,-\frac{u}{\sqrt{-v^2-u^2+1}}]$

(normal) $[\frac{u}{\sqrt{-v^2-u^2+1}},\frac{v}{\sqrt{-v^2-u^2+1}},1]$

--> `/* Cálculo de la función integrando */`

```
(%i47)  f:[x(u,v)*z(u,v)^2,x(u,v)^2*y(u,v)-z(u,v)^3,
        2*x(u,v)*y(u,v)+y(u,v)^2*z(u,v)];
        int:f.normal;
```

(f) $[u\left(-v^2-u^2+1\right), u^2\,v-\left(-v^2-u^2+1\right)^{3/2}, v^2\sqrt{-v^2-u^2+1}+2\,u\,v]$

(int) $\dfrac{v\left(u^2\,v-\left(-v^2-u^2+1\right)^{3/2}\right)}{\sqrt{-v^2-u^2+1}}+v^2\sqrt{-v^2-u^2+1}+u^2\sqrt{-v^2-u^2+1}+2\,u\,v$

```
(%i48)  int:ratsimp(int);
```

(int) $\dfrac{-v^4+\sqrt{-v^2-u^2+1}\left(v^3+\left(u^2+2\,u-1\right)v\right)+\left(1-u^2\right)v^2-u^4+u^2}{\sqrt{-v^2-u^2+1}}$

```
-->     /* Cálculo de la integral */
(%i49)  assume(1-u^2-v^2>0);
```

(%o49) $[-v^2-u^2+1>0]$

```
(%i50)  assume(1-u^2>0);
```

(%o50) $[u^2<1]$

```
(%i51)  iterada1:integrate(int,v,-sqrt(1-u^2),sqrt(1-u^2));
```

(iterada1) $-\dfrac{7\,\pi\,u^4-6\,\pi\,u^2-\pi}{8}$

```
(%i52)  integrate(iterada1, u, -1, 1);
```

(%o52) $\dfrac{2\,\pi}{5}$

```
-->     /* b) En coordenadas esféricas */

-->     /* Ecuaciones paramétricas */
(%i55)  x(u,v):=cos(u)*sin(v);
        y(u,v):=sin(u)*sin(v);
        z(u,v):=cos(v);
```

(%o53) $x(u,v):=\cos(u)\sin(v)$

(%o54) $y(u,v):=\sin(u)\sin(v)$

(%o55) $z(u,v):=\cos(v)$

```
-->     /* Cálculo de la normal */
```

```
(%i61)  t(u,v):=[x(u,v),y(u,v),z(u,v)];
        tu:diff(t(u,v),u);
        tv:diff(t(u,v),v);
        n:tu~tv$
        normal:express(n)$
        normal:trigsimp(normal);
```

(%o56) $t(u,v):=[x(u,v),y(u,v),z(u,v)]$

(tu) $[-\sin(u)\sin(v),\cos(u)\sin(v),0]$

(tv) $[\cos(u)\cos(v),\sin(u)\cos(v),-\sin(v)]$

(normal) $[-\cos(u)\sin(v)^2,-\sin(u)\sin(v)^2,-\cos(v)\sin(v)]$

```
-->      /* Cálculo de la función integrando */
```

```
(%i64)  f:[x(u,v)*z(u,v)^2,x(u,v)^2*y(u,v)-z(u,v)^3,
        2*x(u,v)*y(u,v)+y(u,v)^2*z(u,v)]$
        int:f.normal$
        int:trigsimp(int);
```

(int) $\left(\sin(u)^4-\sin(u)^2\right)\sin(v)^5+\left(-\cos(v)^2-2\cos(u)\sin(u)\cos(v)\right)\sin(v)^3+\sin(u)\cos(v)^3\sin(v)^2$

```
-->      /* Cálculo de la integral */
```

```
(%i65)  iterada1:integrate(int,v,0,%pi/2);
```

(iterada1) $\dfrac{16\sin(u)^4-16\sin(u)^2+\left(4-15\cos(u)\right)\sin(u)-4}{30}$

```
(%i66)  integrate(iterada1, u, 0, 2*%pi);
```

(%o66) $-\dfrac{2\pi}{5}$

```
-->      /* Se puede comprobar que, en un caso hemos hallado
          la integral de superficie en la dirección de la normal exterior,
          y en el otro caso en la dirección de la
         normal interior */
```

EJERCICIOS PROPUESTOS

1.

Calcular la integral de superficie.

$$\iint_S (x^2+y^2)dS$$

Siendo S la superficie de la esfera $x^2+y^2+z^2=25$.

Solución: $\mathbf{\dfrac{5000\pi}{3}}$

2.

Calcular el área de la parte del paraboloide:.

$$x^2+y^2+z=5$$

que está por encima del plano $z=1$.

Solución: $\mathbf{36.176903197411}$

3.

Calcular la integral de superficie $\iint_S \overrightarrow{F}.\overrightarrow{dS}$, siendo

$$\overrightarrow{F}=x\overrightarrow{i}+y\overrightarrow{j}$$

y S la porción de superficie cónica $z^2=x^2+y^2$, $0\le z\le 1$.

Solución: $\mathbf{-\dfrac{2\pi}{3}}$

4.

Calcular la integral de superficie $\iint_S ydydz-xdxdz+zdxdy$, siendo S la superficie $x^2+y^2+z^2=1$, $z\ge 0$.

Solución: $\mathbf{-\dfrac{2\pi}{3}}$

5.

Calcular el momento de inercia I_z, respecto del eje OZ, de la superficie esférica $x^2+y^2+z^2=9$. Considérese la densidad superficial $\mu(x,y,z)=x^2+y^2+z^2$.

Solución: $\mathbf{1944\pi}$

6.

Calcular la integral de superficie $\iint_S \overrightarrow{F}\cdot\overrightarrow{dS}$, siendo $\overrightarrow{F}$ el campo vectorial:

$$\overrightarrow{F}=0\,\overrightarrow{i}+0\,\overrightarrow{j}-2\,\overrightarrow{k}$$

y S la superficie esférica $x^2+y^2+z^2=2$, interceptada por el cono $z=\sqrt{x^2+y^2}$ y siendo $z\geqslant 1$.

Solución: $\mathbf{-2\pi}$

7.

Calcular la integral de superficie $\iint_S \overrightarrow{F} \cdot \overrightarrow{dS}$, siendo $\overrightarrow{F}$ el campo vectorial:

$$\overrightarrow{F} = (z+x)\,\overrightarrow{i} + 0\,\overrightarrow{j} - (z+x)\,\overrightarrow{k}$$

y S la superficie esférica $x^2 + y^2 + z = 2$, interceptada por el plano $z = 1$ y siendo $z \geqslant 1$.

Solución: $-\pi$